



Задача А. Киригами

Учитывая, что в результате разрезов получались прямоугольники, то разрез был вдоль одной из сторон.

Пусть исходный прямоугольник имел стороны a , b .




Тогда при разрезаниях получались прямоугольники:



При таком разрезе сумма периметров двух разрезанных прямоугольников

$$2P = 4a + 2b$$

При втором разрезании:



сумма периметров двух разрезанных прямоугольников

$$2R = 2a + 4b$$

Тогда

$$2R + 2P = 6a + 6b,$$

а это в три раза больше, чем искомый периметр.

Пример программы на языке Pascal

```
program kirigami;  
var p, r: integer;  
begin  
  read(p);  
  read(r);  
  write((2 * p + 2 * r) div 3);  
end.
```

Задача В. Непростые числа

Для решения нужно заметить, что число будет иметь 5 делителей в том случае, когда оно является четвертой степенью простого числа. Поэтому решение задачи сводится к поиску всех простых чисел в диапазоне $[1, \sqrt[4]{N}]$

Пример программы на языке Pascal

```
program unsimple;
var n, i, k, count, del: integer;
begin
  read(n);
  if n >= 16 then
    writeln(16);
  k := 3;
  while k * k * k * k <= n do
    begin
      count := 0;
      del := 2;
      while (count = 0) and (del * del <= k) do
        begin
          if k mod del = 0 then
            count := count + 1;
          del := del + 1;
        end;
      if count = 0 then
        writeln(k * k * k * k);
      k := k + 1;
    end;
end.
```

Задача С. Учебный исполнитель

Чтобы найти расстояние от начальной до конечной точки, можно воспользоваться теоремой Пифагора, для чего сначала следует узнать смещение вдоль оси x и вдоль оси y .

Учитывая, что можно поворачиваться на 90 градусов вправо и влево вокруг своей оси, то можно двигаться в четырех направлениях: север, юг, запад, восток. Обозначим их числовыми значениями: 0 – север (положительное направление оси y), 1 – запад (отрицательное направление оси x), 2 – юг (отрицательное направление оси y), 3 – восток (положительное направление оси x). Тогда, при каждом повороте направо, «направление» уменьшается на 1 , в налево – увеличивается. Чтобы не «выйти» за границы допустимых значений направления, будем получать остаток от деления на 4 .

Зная направления, вычисляем смещение по x и y , а затем находим расстояние и округляем его.

Пример программы на языке Pascal

```
program executor;
var s: string;
n, i, x, y: integer;
begin
  readln(s);
  x := 0;
  y := 0;
  n := 0;
  for i := 1 to length(s) do
    begin
      if s[i] = 'L' then
        n := n + 1;
      if s[i] = 'R' then
        n := n - 1;
      n := (n + 4) mod 4;
      if n = 0 then y := y + 1;
      if n = 1 then x := x - 1;
      if n = 2 then y := y - 1;
      if n = 3 then x := x + 1;
    end;
  write(trunc(sqrt(x * x + y * y)));
end.
```

Задача D. Премии к Новому году

Вычислим для начала максимальный размер премии, который мог бы получиться, если оба начальника выделяют наибольшие суммы, а средства делят поровну между всеми работниками.

Проанализируем, сколько всего средств потребуется на премии такой величины для всех работников. Если эта величина меньше $2a$ (это может быть при большом количестве работников и малым размахом между a и b), то тогда обоим начальникам следует выделить по a условных единиц.


Если все-таки сумма финансирования на все премии не меньше $2a$, то сделаем так, чтобы первый руководитель выделил большие средства (т.е. b), а все остальное – второй руководитель (но не меньше, чем a условных единиц).

Пример программы на языке Pascal

```
program prize;
var n, a, b, cnt, x, y: integer;
begin
  read(n);
  read(a);
  read(b);
  cnt := 2 * b div n;
  if cnt * n < 2 * a then
    begin
      x := a;
      y := a;
    end
  else
    begin
      x := b;
      y := cnt * n - b;
      if y < a then
        begin
          x := cnt * n - a;
          y := a;
        end;
    end;
  write(x, ' ', y);
end.
```

Задача Е. Начинающий дизайнер

По условию, клетки, расположенные на диагоналях, выложены образцом «а» (на рисунке выделены красным цветом)



Для ячеек главной диагонали выполняется свойство, что $i = j$ (индексы строки и столбца), а для ячеек побочной диагонали $i + j = n + 1$ (или $i + j = n - 1$, если нумерация строк и столбцов ведется с 0).

Остальные ячейки можно определить по расстоянию до одной из диагоналей (очевидно, что нужно выбрать ближайшую диагональ). Учитывая, что ячейки считаются соседними, если у них есть общая сторона, то расстояние до ближайшей диагонали считается как минимальное из расстояний по горизонтали и по вертикали.

Определив расстояние, восстанавливаем нужный код символа (код символа «а» плюс найденное расстояние), а по нему – сам символ.

Пример программы на языке Pascal

```
Program designer;
var n, i, j, r: integer;
a: array[1..100, 1..100] of char;
begin
  read(n);
  for i := 1 to n do
    for j := 1 to n do
      begin
        r := min(abs(i - j), abs(n + 1 - i - j));
        a[i, j] := chr(ord('a') + r mod 26);
      end;
  for i := 1 to n do
    begin
      for j := 1 to n do
        write(a[i, j]);
      writeln;
    end;
end.
```